Beyond the Black Hole Horizon

Max Coplan, Dr. James Overduin
Towson University

Black Holes

- Are real!


Barry C. Barish (Caltech)
Kip S. Thorne (Caltech)
Rainer Weiss (MIT)

Nobel Prize

Inside the Horizon

- Black-hole interiors are the only place in nature we **can never observe** — by definition!
- The only way to “see” beyond the event horizon is through mathematics
- Existing visualizations in the realistic case (mass and spin but no electric charge) use special coordinates known as *Boyer-Lindquist* coordinates
- This brings out the singularity (ring) and horizon structure (spherical inner horizon, ellipsoidal ergosphere):



Invariants

- Simplicity of the preceding picture is due to the choice in coordinates
- But coordinate-dependent quantities can be misleading! Consider Greenland vs. U.S.A. on different projections
- To be sure of drawing true conclusions, we need to express results in terms of invariants: quantities whose value is the same regardless of coordinates

Objective

- We focus on the **curvature of spacetime** inside the black hole horizon
- Mathematicians have proved that this curvature is characterized by at most **seventeen invariants** for the most general possible black holes (those with mass, spin, and charge)
- We use a powerful symbolic computational tool to calculate and plot all seventeen of these curvature invariants for the first time
- The results differ dramatically from the simple picture above!



General Relativity

- In GR (Einstein, 1916), **gravity = shape of spacetime**
- Shape of spacetime is described mathematically by the *metric tensor*, a generalization of Pythagoras' theorem

Flat or Minkowski spacetime (no gravity): $g_{ij} = \begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$

Schwarzschild black hole (mass M but no charge or spin): $g_{ij} = \begin{pmatrix} -\left(1 - \frac{2GM}{c^2 r}\right)^2 & 0 & 0 & 0 \\ 0 & \frac{1}{1-\frac{2GM}{c^2 r}} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2 \mathrm{Sin}[\theta]^2 \end{pmatrix}$

- All physical quantities are then constructed out of the metric and its first and second derivatives with respect to space and time:

Metric tensor: g_{ij} (simplest examples above) (2D matrix, $4^2 = 16$ terms)

Christoffel symbol: $\Gamma^i{}_{jk} = \frac{1}{2} g^{il} (\partial_j g_{lk} + \partial_k g_{lj} - \partial_l g_{jk})$ (3D matrix, $4^3 = 64$ terms)

Riemann tensor: $R^i{}_{jkl} = \partial_k \Gamma^i{}_{jl} - \partial_l \Gamma^i{}_{jk} + \Gamma^o{}_{jl} \Gamma^i{}_{ok} - \Gamma^o{}_{jk} \Gamma^i{}_{ol}$ (4D matrix, $4^4 = 256$ terms)

Fully covariant and contravariant forms: $R_{ijkl} = g_{io} R^o{}_{jkl}$ and $R^{ijkl} = g^{jo} g^{kp} g^{ls} R^i{}_{ops}$ (summation convention)

Ricci tensor: $R_{ij} = R^k{}_{ikj}$ (contraction)

Implementation

- Two main challenges:
  - *Translating* tensor language of general relativity to computational syntax (easy)
  - *Simplifying* the resulting expressions (hard)

Translation

- Our inspiration is a Mathematica code in James Hartle's Gravity (Addison-Wesley, 2003)
- Given a metric, our code computes the inverse metric, Christoffel symbols, Riemann and Ricci Tensors

```
Clear[n, coord, r, θ, ϕ, t, m, Q, a, metric, inversemetric, christoffel, riemann, riemannDown, ricci, listricci,
 scalar, ricciSimp, listricciSimp]
```

Coordinates

```
n = 4; coord = {r, θ, ϕ, t};
```

Metric

```
metric = {{(r^2 + (a Cos[θ])^2)/(r^2 + a^2 + Q^2 - 2 m r), 0, 0, 0}, {0, r^2 + (a Cos[θ])^2, 0, 0},
  {0, 0, (r^2 + a^2 - (a^2 (Q^2 - 2 m r) (Sin[θ])^2)/(r^2 + (a Cos[θ])^2)) (Sin[θ])^2, (a (Q^2 - 2 m r) (Sin[θ])^2)/(r^2 + (a Cos[θ])^2)},
  {0, 0, (a (Q^2 - 2 m r) (Sin[θ])^2)/(r^2 + (a Cos[θ])^2), -(1 + (Q^2 - 2 m r)/(r^2 + (a Cos[θ])^2))}};
```

Note: we reserve a for angular momentum, g for the determinant of the metric, m for mass, n for dimensionality, Q for charge, r for (Boyer-Lindquist) radial distance, s for proper time and t for time. Variables i, j, k, l, o, p, u and v may be used for dummy indices.

Inverse metric

```
inversemetric = Inverse[metric];
```

Christoffel symbols

Obtained from $\Gamma^i{}_{jk} = \frac{1}{2} g^{il} (\partial_j g_{lk} + \partial_k g_{lj} - \partial_l g_{jk})$:

```
christoffel := christoffel = Table[1/2 Sum[inversemetric[[i, l]] ×
    (D[metric[[l, k]], coord[[j]]] +
      D[metric[[l, j]], coord[[k]]] - D[metric[[j, k]], coord[[l]]]), {l, 1, n}],
  {i, 1, n}, {j, 1, n}, {k, 1, n}]
```

Riemann tensor

Calculated from $R^i{}_{jkl} = \partial_k \Gamma^i{}_{jl} - \partial_l \Gamma^i{}_{jk} + \Gamma^o{}_{jl} \Gamma^i{}_{ok} - \Gamma^o{}_{jk} \Gamma^i{}_{ol}$:

```
riemann := riemann = Table[
   D[christoffel[[i, j, l]], coord[[k]] ] - D[christoffel[[i, j, k]], coord[[l]] ] +
    Sum[christoffel[[o, j, l]] christoffel[[i, k, o]] - christoffel[[o, j, k]] christoffel[[i, l, o]],
     {o, 1, n}],
   {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]
```

We will also need the fully covariant form (all indices down) of the Riemann tensor, $R_{ijkl} = g_{io} R^o{}_{jkl}$:

```
riemannDown := riemannDown = Table[Sum[metric[[i, o]] riemann[[o, j, k, l]], {o, 1, n}], {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}]
```

Ricci tensor

Obtained by contracting on the first and third indices of Riemann, $R_{ij} = R^k{}_{ikj}$:

```
ricci := ricci = Table[Sum[riemann[[k, i, k, j]], {k, 1, n}], {i, 1, n}, {j, 1, n}]
```

Check nonzero components:

```
listricci := Table[If[UnsameQ[ricci[[j, l]], 0], {ToString[R[j, l]], ricci[[j, l]]}] , {j, 1, n}, {l, 1, j}]

TableForm[Partition[DeleteCases[Flatten[listricci], Null], 2], TableSpacing → {2, 2}]
```

Ricci scalar

We'll also need the Ricci scalar, defined by $R = g^{ij} R_{ij}$:

```
scalar = Sum[inversemetric[[i, j]] ricci[[i, j]], {i, 1, n}, {j, 1, n}]
```

(This is also the simplest of the seventeen curvature invariants, known as the *Ricci invariant*)

Issue: Pages and pages of results



Simplification

- The more difficult challenge is to simplify the resulting expressions! For example, messy combinations of $\sin(\theta)$ and $\cos(2\theta)$ terms in R_{ij} above --- we would like to express all results in terms of powers of a single trigonometric function
- To do this we constructed a "trig simplifier":

Trig Simplifier:

```
$TrigFns = {Sin, Cos, Tan, Csc, Sec, Cot};
(WRules = $TrigFns == (Through[$TrigFns[x]] /. x → 2 ArcTan[t] // TrigExpand // Together) // Thread);

invWRules = #[[1]] → Solve[#, t, Reals] & /@ WRules;

convert[expr_, (trig : Alternatives @@ $TrigFns)[x_]] := Block[{temp, t}, temp = expr /. x → 2 ArcTan[t] // TrigExpand // Factor;
  temp = temp /. (trig /. invWRules) // Union // FullSimplify;
  Or @@ temp /. trig → HoldForm[trig][x] /. ConditionalExpression → (#1 &) // FullSimplify]
```

Simplification results

A known result 👍



Example: Weyl Invariants





2
0
-2





Physical Implications

Conclusions

- Curvature inside real black holes is (1) both positive and negative, (2) definitely not constant, and (3) ridiculously complex!
- The interiors of black holes are much more complex and beautiful than we ever imagined
- The Weyl invariants are especially contorted. Interesting since these describe curvature associated with gravitational waves
- Spacetime is mostly warped ("gravito-electric" dominated) in regions of positive curvature, but mostly twisted/dragged ("gravito-magnetic" dominated) in regions of negative curvature
- This may explain certain puzzles in astrophysics
- More remains to be understood about the physical importance of these invariants
- But before that, it is necessary to find the invariants, and that is what we have done here!